


Welcome!


Screencasts


Quiz 1 and Assignment 1

✔ Quiz: Quiz 1
5 questions

 **Reading: Quiz 1**
solutions and
explanations
10 min

 Reading: Assignment 1
2h

 **Video:** Assignment 1
preview and instructions
2 min

 **Video:** More about the OFFSET function
7 min

 Discussion Prompt:
Assignment 1 discussion
10 min

 Quiz: Assignment 1
submission
1 question

 Reading: Week 2 Excel files
10 min

Quiz 1 solutions and explanations

Quiz 1 Solutions & Explanations

Everyday Excel, Part 2

Charlie Nuttelman

Hello there! This document is meant to provide clear explanations for the Quiz 1 questions (not the in-video quizzes since they have explanations already). I do NOT provide feedback during the quiz (like I do for the screencasts) because a learner could just guess, obtain the correct answers, then put them back into the quiz and get 100%!

This document is purely for you to learn more and to correct your misconceptions about the material. If you view this document soon after you take the quiz to see why you missed a certain question, it will serve as a great learning tool!

PLEASE DO NOT SHARE THIS DOCUMENT WITH ANYONE! Using this document to complete Quiz 1 is a violation of Coursera's Honor Code (a.k.a. cheating).

NOTE that the order of the answers on Coursera are random and likely different from the order shown here (in general but not always. I like to start with the correct answers followed by the incorrect ones).

Question 1:

Which of the following conditional formatting rules was applied to cells A1:C3 to result in the formatted range shown below?

	A	B	C	D	E
1	7	4	3		
2	12	9	1		10
3	-4	8	2		
4					



	A	B	C	D	E
1	7	4	3		
2	12	9	1		10
3	-4	8	2		

Correct answer:

Edit the Rule Description:

Format values where this formula is true:

=OR(\$A1>\$E\$2,\$B1>\$E\$2,\$C1>\$E\$2)

Preview:

AaBbCcYvZz

Format...

Remember, if we use a formula in conditional formatting, we will format only when the conditional statement/formula evaluates to **TRUE**. If any (that's what the OR statement does) of the cells **A1**, **B1**, or **C1** is greater than 10 then we will format the row. We use **\$A1**, **\$B1**, and **\$C1** so that we will highlight the entire row.

A good way to set up the formula in the conditional formatting tool (remember, it only applies to the upper leftmost cell of the range that you are applying the rule to) is to set up another part of the worksheet. In cell **A5** I've entered the formula **"=OR(\$A1:\$F\$2,\$B1:\$E\$2,\$C1:\$E\$2)"** and I've copied/pasted that formula into the entire range **A5:C7**.

	A	B	C	D	E
1	7	4	3		
2	12	9	1		10
3	-4	8	2		
4					
5	FALSE	FALSE	FALSE		
6	TRUE	TRUE	TRUE		
7	FALSE	FALSE	FALSE		
8					

If I don't use the mixed references for **SA1**, **SB1**, and **SC1** ("=OR(A1>\$E\$2,B1>\$E\$2,C1>\$E\$2)" in cell **A5**) then we would get the following:

	A	B	C	D	E
1	7	4	3		
2	12	9	1		10
3	-4	8	2		
4					
5	FALSE	FALSE	=OR(C1>\$E\$2,D1>\$E\$2,E1>\$E\$2)		
6	TRUE	FALSE	FALSE		

6	TRUE	FALSE	FALSE			
7	FALSE	FALSE	FALSE			
8						

Note that the formula in cell **C5** is not looking at the entire row 1 of the array of interest. And, for example, row 2 is not comprised of all **TRUE** values, so only cell **A6** will be conditionally formatted.

Question 2:

Which of the following formulas could be placed into cell **C4** and dragged down through **C4:C16** and would provide the cumulative (starting from the top) number of errors that have occurred in each **Process**?

	A	B	C
1	PROCESS ERRORS		
2			
3	Process	Date and Time	# Errors
4	10-A	1/31/20 12:39 PM	1
5	11-B	1/31/20 2:44 PM	1
6	09-A	1/31/20 10:01 PM	1
7	10-A	1/31/20 11:26 PM	2
8	09-A	2/1/20 3:36 AM	2
9	07-C	2/1/20 3:50 AM	1
10	11-C	2/1/20 4:20 AM	1
11	09-A	2/1/20 5:22 AM	3
12	11-B	2/1/20 7:13 AM	2
13	04-B	2/1/20 10:36 AM	1
14	06-D	2/1/20 5:00 PM	1
15	13-A	2/1/20 8:59 PM	1
16	11-B	2/1/20 11:25 PM	3

Correct answer: =COUNTIF(\$A\$4:A4,A4)

Explanation: We can use the **COUNTIF** function to count the number of items in the range in the first argument that are equal to the second argument. If we are counting up to the current row from the first row, we must "pin" the starting point with a **\$A\$4** here. This will only count from cell **\$A\$4** to whichever row we are currently in.

For example, when we drag this formula down into cell **C7**, the formula in cell **C7** will be: "=COUNTIF(\$A\$4:A7,A7)", which will count the number of occurrences of "10-A" from cells **\$A\$4** to **A7**, which is 2.

Question 3:

The worksheet below shows the **Log in** and **Log out** times for 10 different people to a specific building over the course of the day. Which of the following formulas has been placed in cell **F4** that will calculate the total number of people in the building at the **Time** entered in cell **F3**? (Ctrl-Shift-Enter was pressed on versions other than Office 365.)

	A	B	C	D	E	F
1	Name	Log in	Log out			
2	Jimmy	9:10 AM	12:32 PM			
3	Jenny	9:34 AM	9:56 AM		Time:	12:00 PM
4	Sophia	10:07 AM	12:54 PM		Total in building:	3
5	Cyndy	10:09 AM	11:22 AM			
6	Mike	11:10 AM	3:31 PM			
7	Nancy	12:21 PM	1:09 PM			
8	Terry	12:45 PM	2:28 PM			
9	Henry	1:12 PM	1:27 PM			
10	Wally	1:23 PM	4:43 PM			
11	Otto	2:38 PM	3:33 PM			

Correct answer: =SUM((B2:B11<=F\$3)*(C2:C11>=F\$3))

Explanation: Let's start with what "**B2:B11<=F\$3**" does. This will output a column vector of **TRUE/FALSE** values - **TRUE** if the items in **B2:B11** (**Log in** time) are less than or equal to the time in cell **F3** (12:00 PM) and **FALSE** if the time is greater than 12:00 PM. The "**C2:C11>=F\$3**" term will do something similar - it will output a column of **TRUE/FALSE** values depending on whether the corresponding element of **C2:C11** is later than 12:00 PM (**TRUE**) or earlier than 12:00 PM (**FALSE**). In order for a person to have been in the building at the time in cell **F3**, both of the **TRUE/FALSE** options above must be **TRUE**. In Excel you can multiply booleans (variables that can take on **TRUE** or **FALSE** values). A **TRUE*TRUE = 1** and **TRUE*FALSE, FALSE*TRUE, and FALSE*FALSE** are all = 0.

So, when we multiply (**B2:B11<=F\$3**) by (**C2:C11>=F\$3**), we will get an array (column vector) of 1's and 0's, and if we sum that array this will tell us the total number of people in the building at 12:00 PM (or whatever time is in cell **F3**).

Question 4:

We wish to determine the rows and columns that the 7 is found in an array. In cell **E1**, we've placed the formula shown, and we've copied/pasted this formula into cells **E2** and **E3**. Thus, the values in cells **E1:E3** will tell us which row the 7 is found in.

A: What formula has been placed in cell **C5**?

B: What formula has been placed in cell **C6**?

	A	B	C	D	E	F
1	3	7	5		=COUNTIF(A1:C1,7)	
2	1	8	6		0	
3	9	4	2		0	
4						
5		Row:	1			
6		Col:	2			
7						

Correct answer:

A: =MATCH(1,E1:E3,0)

B: =MATCH(7,OFFSET(A1,C5-1,0,1,3),0)

Explanation: The formulas in cells **E1:E3** will count the number of occurrences of 7 in each row. We are assuming that there is only one 7 in the array. Now that we know the row that the 7 is in (row 1 in our example), we need to search through that particular row for the 7.

This would be easy if it were ALWAYS in row 1; we could just write the formula "=MATCH(7,A1:C1,0)" to determine the column of the 7 in row 1. However, the 1 in column **E1:E3** is not always going to be in the first row. We need to somehow implement the **Row** number in our calculation.

A: To determine the **Row** number of the 7, we can use the **MATCH** function in cell **C5**: "=MATCH(1,E1:E3,0)", which will search for a 1 in cells **E1:E3**, and therefore tell us which row contains the 7.

B: To determine the **Col** number of the 7, we need to use the **Row** number in combination with the **OFFSET** function. If the 7 is found in **Row 1**, then we'll use the **MATCH** function in cells **A1:C1**; if the 7 is found in **Row 2**, then we'll use the **MATCH** function in cells **A2:C2**; and if the 7 is found in **Row 3**, then we'll use the **MATCH** function in cells **A3:C3**. We can use "=OFFSET(A1,C5-1,0,1,3)" to output the row depending on the **Row** number in cell **C5**. We must subtract 1 from **C5** because we are offsetting from the first column, but we want to use the first column as the leftmost cell of each row.

Question 5:

We wish to determine which of the items on **List 2** are also on **List 1**. What formula can be placed in cell **D2** and dragged down through **D2:D15** that will output a "YES" if the corresponding item in column **C** is found on **List 1** (column **A**) and will be blank if the item is not found on **List 1**?

	A	B	C	D
1	List 1		List 2	On List 1?
2	38-392		43-595	
3	85-894		27-839	YES
4	63-612		77-265	
5	33-238		48-226	YES
6	96-298		88-607	
7	45-556		62-288	
8	55-795		27-666	
9	48-226		87-633	
10	45-779		25-585	YES
11	60-995			
12	25-585			
13	68-133			
14	52-891			
15	27-839			

Correct answer: =IF(COUNTIF(\$A\$2:\$A\$15,C2),"YES","")

Explanation: We can use the **COUNTIF** function in cell **D2** to search through to see if the value to the left of it (in cell **C2**) is found in column **A** (**\$A\$2:\$A\$15**) using "=COUNTIF(\$A\$2:\$A\$15,C2)". When we drag this formula down through cell **D10**, this will result in a bunch of 1's and 0's (1 if it was found in column A on **List 1**; 0 if it was not found on **List 1**). But, if we wish to convert those 1's to a "YES" (and blank where we have a 0), we can just use the **IF** function: =IF(COUNTIF(\$A\$2:\$A\$15,C2),"YES","").

Mark as completed